

## Taylor's Error

**Taylor's Theorem.** Suppose  $f$  has  $n + 1$  derivatives for every  $x$  for  $|x - a| \leq d$  (or  $a - d \leq x \leq a + d$ ). Then

$$f(x) = P_n(x) + E_n(x)$$

on  $|x - a| \leq d$  (or  $a - d \leq x \leq a + d$ ) where

$$P_n(x) = \sum_{k=0}^n \frac{1}{k!} f^{(k)}(a) (x - a)^k.$$

Also, if  $|f^{(n+1)}| \leq M$  for  $|x - a| \leq d$  (or  $a - d \leq x \leq a + d$ ), then

$$|E_n(x)| = |f(x) - P_n(x)| \leq \frac{M}{(n+1)!} |x - a|^{(n+1)}$$

for  $|x - a| \leq d$  (or  $a - d \leq x \leq a + d$ ).

```
> restart:with(plots):
```

```
Warning, the name changecoords has been redefined
```

We first find the 4th degree Taylor polynomial  $P_4(x)$  for the function  $f(x) = x e^{(x^2)}$  about  $x = 0$ .

```
> f:=x*exp(x^2);
```

$$f := x e^{(x^2)}$$

```
> T[4]:=taylor(f,x=0,5);
```


```
> P[4]:=convert(T[4],polynom);
```

$$T_4 := x + x^3 + O(x^5)$$

$$P_4 := x + x^3$$

We observe that the third and fourth Taylor polynomials are the same here. We want to find an upper bound for the error  $|E_4(x)| = |f(x) - P_4(x)|$  for  $-0.4 \leq x \leq 0.4$ . Let's first look at the two graphs, with the Taylor polynomial in blue.

```
> plot([f,P[4]],x=-0.4..0.4,color=[red,blue]);
```




The graphs look pretty close. Now, to find a maximum for  $|E_4(x)|$  on  $-0.4 \leq x \leq 0.4$ , we need to find the fifth derivative.

```
> f5:=diff(f,x$5);
```

$$f5 := 60 e^{(x^2)} + 360 x^2 e^{(x^2)} + 240 x^4 e^{(x^2)} + 32 x^6 e^{(x^2)}$$

We now need to find a number greater than the maximum value of the absolute value of this fifth derivative over the interval. We prefer that this number be chosen as small as possible. We graph the fifth derivative over the interval.

```
> plot(f5,x=-0.4..0.4);
```



From the graph it is clear that the absolute value of the fifth derivative never exceeds 150 over our interval, so we choose that number for  $M$ .

```
> M:=150;
```

```
M := 150
```


Now we can find an upper bound for the error over the interval using the error formula.

```
> maxerror:=M/5!*0.4^5;
```

```
maxerror := .01280000000
```

Graphing the error  $E_4(x) = f(x) - P_4(x)$  below, we see the error is actually between  $-.006$  and  $.006$ , certainly smaller than the maximum we computed.

```
> plot(f-P[4],x=-0.4..0.4);
```



Now suppose that we wish to find the Taylor polynomial that will approximate  $f$  over this same interval with a maximum error of 0.00001. We need to find the degree  $n$ . We use a [while](#) loop indexed on  $n$ . We also use the [maximize](#) command.

```

> n:=0;
  while
  1/(n+1)!*maximize(abs(diff(f,x$(n+1))),x=-0.4..0.4)*0.4^(n+1)>0.00001
  do
  n:=n+1
  od;

      n := 0
      n := 1
      n := 2
      n := 3
      n := 4
      n := 5
      n := 6
      n := 7
      n := 8
      n := 9

```

For the  $n$  just calculated, we find the Taylor polynomial  $P_n(x)$ .

```

> T[n]:=taylor(f,x=0,n+1);
> P[n]:=convert(T[n],polynom);


```

$$T_9 := x + x^3 + \frac{1}{2}x^5 + \frac{1}{6}x^7 + \frac{1}{24}x^9 + O(x^{11})$$

$$P_9 := x + x^3 + \frac{1}{2}x^5 + \frac{1}{6}x^7 + \frac{1}{24}x^9$$

Finally, we graph the error  $E_n(x) = f(x) - P_n(x)$ .

```
> plot(f-P[n],x=-0.4..0.4);
```



We see that the error is less than our requirement.

```
>
```

```
>
```